# NAG Library Routine Document

# F07BGF (DGBCON)

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

## 1 Purpose

F07BGF (DGBCON) estimates the condition number of a real band matrix $A$, where $A$ has been factorized by F07BDF (DGBTRF).

## 2 Specification

```
SUBROUTINE F07BGF (NORM, N, KL, KU, AB, LDAB, IPIV, ANORM, RCOND, WORK,         &
                   IWORK, INFO)

INTEGER             N, KL, KU, LDAB, IPIV(*), IWORK(N), INFO
REAL (KIND=nag_wp) AB(LDAB,*), ANORM, RCOND, WORK(3*N)
CHARACTER(1)        NORM
```

The routine may be called by its LAPACK name ***dgbcon***.

## 3 Description

F07BGF (DGBCON) estimates the condition number of a real band matrix $A$, in either the 1-norm or the $\infty$-norm:

$$\kappa_1(A) = \|A\|_1 \left\|A^{-1}\right\|_1 \qquad \text{or} \qquad \kappa_\infty(A) = \|A\|_\infty \left\|A^{-1}\right\|_\infty.$$

Note that $\kappa_\infty(A) = \kappa_1\left(A^{\mathrm{T}}\right)$.

Because the condition number is infinite if $A$ is singular, the routine actually returns an estimate of the **reciprocal** of the condition number.

The routine should be preceded by a call to F06RBF to compute $\|A\|_1$ or $\|A\|_\infty$, and a call to F07BDF (DGBTRF) to compute the $LU$ factorization of $A$. The routine then uses Higham's implementation of Hager's method (see Higham (1988)) to estimate $\left\|A^{-1}\right\|_1$ or $\left\|A^{-1}\right\|_\infty$.

## 4 References

Higham N J (1988) FORTRAN codes for estimating the one-norm of a real or complex matrix, with applications to condition estimation *ACM Trans. Math. Software* **14** 381–396

## 5 Parameters

1: NORM – CHARACTER(1) *Input*

*On entry*: indicates whether $\kappa_1(A)$ or $\kappa_\infty(A)$ is estimated.

NORM = '1' or 'O'
    $\kappa_1(A)$ is estimated.

NORM = 'I'
    $\kappa_\infty(A)$ is estimated.

*Constraint*: NORM = '1', 'O' or 'I'.

2: N – INTEGER *Input*

*On entry*: $n$, the order of the matrix $A$.

*Constraint*: $N \geq 0$.

3: KL – INTEGER *Input*

*On entry*: $k_l$, the number of subdiagonals within the band of the matrix $A$.

*Constraint*: $KL \geq 0$.

4: KU – INTEGER *Input*

*On entry*: $k_u$, the number of superdiagonals within the band of the matrix $A$.

*Constraint*: $KU \geq 0$.

5: AB(LDAB,$*$) – REAL (KIND=nag_wp) array *Input*

**Note**: the second dimension of the array AB must be at least $\max(1, N)$.

*On entry*: the $LU$ factorization of $A$, as returned by F07BDF (DGBTRF).

6: LDAB – INTEGER *Input*

*On entry*: the first dimension of the array AB as declared in the (sub)program from which F07BGF (DGBCON) is called.

*Constraint*: $LDAB \geq 2 \times KL + KU + 1$.

7: IPIV($*$) – INTEGER array *Input*

**Note**: the dimension of the array IPIV must be at least $\max(1, N)$.

*On entry*: the pivot indices, as returned by F07BDF (DGBTRF).

8: ANORM – REAL (KIND=nag_wp) *Input*

*On entry*: if NORM = '1' or 'O', the 1-norm of the **original** matrix $A$.

If NORM = 'I', the $\infty$-norm of the **original** matrix $A$.

ANORM may be computed by calling F06RBF with the same value for the parameter NORM.

ANORM must be computed either **before** calling F07BDF (DGBTRF) or else from a **copy** of the original matrix $A$ (see Section 9).

*Constraint*: $ANORM \geq 0.0$.

9: RCOND – REAL (KIND=nag_wp) *Output*

*On exit*: an estimate of the reciprocal of the condition number of $A$. RCOND is set to zero if exact singularity is detected or the estimate underflows. If RCOND is less than ***machine precision***, $A$ is singular to working precision.

10: WORK($3 \times N$) – REAL (KIND=nag_wp) array *Workspace*

11: IWORK(N) – INTEGER array *Workspace*

12: INFO – INTEGER *Output*

*On exit*: INFO = 0 unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO $< 0$

If INFO $= -i$, the $i$th parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

## 7 Accuracy

The computed estimate RCOND is never less than the true value $\rho$, and in practice is nearly always less than $10\rho$, although examples can be constructed where RCOND is much larger.

## 8 Further Comments

A call to F07BGF (DGBCON) involves solving a number of systems of linear equations of the form $Ax = b$ or $A^{\mathrm{T}}x = b$; the number is usually 4 or 5 and never more than 11. Each solution involves approximately $2n(2k_l + k_u)$ floating point operations (assuming $n \gg k_l$ and $n \gg k_u$) but takes considerably longer than a call to F07BEF (DGBTRS) with one right-hand side, because extra care is taken to avoid overflow when $A$ is approximately singular.

The complex analogue of this routine is F07BUF (ZGBCON).

## 9 Example

This example estimates the condition number in the 1-norm of the matrix $A$, where

$$A = \begin{pmatrix} -0.23 & 2.54 & -3.66 & 0.00 \\ -6.98 & 2.46 & -2.73 & -2.13 \\ 0.00 & 2.56 & 2.46 & 4.07 \\ 0.00 & 0.00 & -4.78 & -3.82 \end{pmatrix}.$$

Here $A$ is nonsymmetric and is treated as a band matrix, which must first be factorized by F07BDF (DGBTRF). The true condition number in the 1-norm is 56.40.

### 9.1 Program Text

```
    Program f07bgfe

!     F07BGF Example Program Text

!     Mark 24 Release. NAG Copyright 2012.

!     .. Use Statements ..
      Use nag_library, Only: dgbcon, dgbtrf, dlangb => f06rbf, nag_wp, x02ajf
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: nin = 5, nout = 6
      Character (1), Parameter         :: norm = '1'
!     .. Local Scalars ..
      Real (Kind=nag_wp)               :: anorm, rcond
      Integer                          :: i, info, j, k, kl, ku, ldab, n
!     .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable  :: ab(:,:), work(:)
      Integer, Allocatable             :: ipiv(:), iwork(:)
!     .. Intrinsic Procedures ..
      Intrinsic                        :: max, min
!     .. Executable Statements ..
      Write (nout,*) 'F07BGF Example Program Results'
!     Skip heading in data file
      Read (nin,*)
      Read (nin,*) n, kl, ku
```

```
      ldab = 2*kl + ku + 1
      Allocate (ab(ldab,n),work(3*n),ipiv(n),iwork(n))

!     Read A from data file

      k = kl + ku + 1
      Read (nin,*)((ab(k+i-j,j),j=max(i-kl,1),min(i+ku,n)),i=1,n)

!     Compute norm of A
!     f06rbf is the NAG name equivalent of the LAPACK auxiliary dlangb
      anorm = dlangb(norm,n,kl,ku,ab(kl+1,1),ldab,work)

!     Factorize A

!     The NAG name equivalent of dgbtrf id f07bdf
      Call dgbtrf(n,n,kl,ku,ab,ldab,ipiv,info)

      Write (nout,*)
      If (info==0) Then

!       Estimate condition number

!       The NAG name equivalent of dgbcon is f07bgf
        Call dgbcon(norm,n,kl,ku,ab,ldab,ipiv,anorm,rcond,work,iwork,info)

        If (rcond>=x02ajf()) Then
          Write (nout,99999) 'Estimate of condition number =', &
            1.0_nag_wp/rcond
        Else
          Write (nout,*) 'A is singular to working precision'
        End If
      Else
        Write (nout,*) 'The factor U is singular'
      End If

99999 Format (1X,A,1P,E10.2)
    End Program f07bgfe
```

## 9.2 Program Data

```
F07BGF Example Program Data
  4  1  2                     :Values of N, KL and KU
 -0.23   2.54  -3.66
 -6.98   2.46  -2.73  -2.13
         2.56   2.46   4.07
               -4.78  -3.82   :End of matrix A
```

## 9.3 Program Results

```
 F07BGF Example Program Results

 Estimate of condition number =  5.64E+01
```